# $Id: PKGBUILD 80962 2012-12-07 23:56:44Z bluewind $
# Maintainer: Florian Pritz <flo@xinu.at>
# Contributor: Sven Kauber <celeon@gmail.com>

pkgname=siege
pkgver=2.72
pkgrel=1
arch=('i686' 'x86_64')
license=('GPL')
pkgdesc="An http regression testing and benchmarking utility"
url="http://www.joedog.org/JoeDog/Siege"
depends=('openssl')
options=(!libtool)
source=("http://www.joedog.org/pub/siege/siege-${pkgver}.tar.gz")
backup=(etc/siegerc etc/urls.txt)

build() {
  cd "${srcdir}/$pkgname-$pkgver"
  ./configure --prefix=/usr --sysconfdir=/etc --mandir=/usr/share/man
  make
}

package() {
  cd "${srcdir}/$pkgname-$pkgver"

  install -d -m755 "$pkgdir/etc"
  install -d -m755 "$pkgdir/usr/share"
  make DESTDIR="$pkgdir" sysconfdir="$pkgdir/etc" install
  sed -i 's/^# file =$/file = \/etc\/urls.txt/' "$pkgdir/etc/siegerc"
}

md5sums=('6bd0b1dca6b95717e23a6bade9a0a1f4')
